32 The angle between two celestial objects

Sometimes it is of interest to know what is the angle between two objects in the sky, and this can be calculated very easily provided their equatorial coordinates (α, δ) or ecliptic coordinates (λ, β) are known. The formula is:

$$\cos d = \sin\delta_1 \sin\delta_2 + \cos\delta_1 \cos\delta_2 \cos(\alpha_1 - \alpha_2)$$

or

$$\cos d = \sin\beta_1 \sin\beta_2 + \cos\beta_1 \cos\beta_2 \cos(\lambda_1 - \lambda_2),$$

where d is the angle between the objects whose coordinates are α_1, δ_1 (or λ_1, β_1) and α_2, δ_2 (or λ_2, β_2). These formulas are exact and mathematically correct for any values of α, δ or λ, β. However, when d becomes either very small, or close to 180°, your calculator may not have enough precision to return the correct answer, in which case a better expression is

$$d = \sqrt{(\cos\delta \times \Delta\alpha)^2 + \Delta\delta^2}$$

or

$$d = \sqrt{(\cos\beta \times \Delta\lambda)^2 + \Delta\beta^2},$$

where $\Delta\alpha$, $\Delta\delta$ (or $\Delta\lambda$, $\Delta\beta$) are the differences in the two coordinates (i.e. $\Delta\alpha = \alpha_1 - \alpha_2$, etc.). These expressions may be used for values of d within about 10 arcmin of 0° or 180°. Both $\Delta\alpha$ ($\Delta\lambda$) and $\Delta\delta$ ($\Delta\beta$) must be expressed in the same units (e.g. arcseconds) and d will then be returned in those units.

For example, what is the angular distance between the stars β Orionis (α = 5h 13m 31.7s; $\delta = -8° 13' 30''$) and α Canis Majoris (α = 6h 44m 13.4s; $\delta = -16° 41' 11''$)?

	Method	*Example*		
1.	Convert both sets of coordinates to decimal form (§§7 and 21).	α_1	=	5.225 472 hours
		δ_1	=	−8.225 000 degrees
		α_2	=	6.737 056 hours
		δ_2	=	−16.686 389 degrees
2.	Find $\alpha_1 - \alpha_2$, and convert to degrees by multiplying by 15 (§22).	$\alpha_1 - \alpha_2$	=	−1.511 583 hours
			=	−22.673 750 degrees
3.	Calculate $\cos d = \sin\delta_1 \sin\delta_2 + \cos\delta_1 \cos\delta_2 \cos(\alpha_1 - \alpha_2)$.	$\cos d$	=	0.915 846
4.	Take inverse cos to find d. Convert to minutes and seconds form if required (§21).	d	=	23.673 850 degrees
			=	**23° 40′ 26″**

The spreadsheet, labelled Angle, is shown in Figure 34. It can use coordinates expressed either in equatorial or ecliptic form, specified via a switch in cell C15. Set this to H (as here) if the coordinates are α, δ (i.e. α is in Hours, minutes and seconds) or D if the coordinates are λ, β (i.e. λ is in Degrees, minutes and seconds). The corresponding spreadsheet function is also called Angle, and it takes the same 13 arguments as entered in the spreadsheet in cells C3 to C15, i.e. the right ascension/longitude of the first object expressed in hours/degrees, minutes, seconds, the declination/latitude of the first object expressed in degrees, minutes and seconds, the same again for the second object, and finally the character H or D specifying the coordinate format.

	A	B	C	D	E	F	G	H	I
1	**The angle between two objects**								
2									
3	*Input*	**RA/long 1 (hour/deg)**	**5**			*Output*	**angle (deg)**	**23**	=DDDeg(C29)
4		**RA/long 1 (min)**	**13**				**angle(min)**	**40**	=DDMin(C29)
5		**RA/long 1 (sec)**	**31.7**				**angle(sec)**	**25.86**	=DDSec(C29)
6		**dec/lat 1 (deg)**	**-8**						
7		**dec/lat 1 (min)**	**13**						
8		**dec/lat 1 (sec)**	**30**						
9		**RA/long 2 (hour/deg)**	**6**						
10		**RA/long 2 (min)**	**44**						
11		**RA/long 2 (sec)**	**13.4**						
12		**dec/lat 2 (deg)**	**-16**						
13		**dec/lat 2 (min)**	**41**						
14		**dec/lat 2 (sec)**	**11**						
15		**Hour/degree [H or D]**	**H**						
16									
17	1	RA/long 1 (decimal)	5.225472222	=IF(C15="H",HMSDH(C3,C4,C5),DMSDD(C3,C4,C5))					
18	2	RA/long 1 (deg)	78.38208333	=IF(C15="H",DHDD(C17),C17)					
19	3	RA/long 1 (rad)	1.368025429	=RADIANS(C18)					
20	4	dec/lat 1 (deg)	-8.225	=DMSDD(C6,C7,C8)					
21	5	dec/lat 1 (rad)	-0.143553331	=RADIANS(C20)					
22	6	RA/long 2 (decimal)	6.737055556	=IF(C15="H",HMSDH(C9,C10,C11),DMSDD(C9,C10,C11))					
23	7	RA/long 2 (deg)	101.0558333	=IF(C15="H",DHDD(C22),C22)					
24	8	RA/long 2 (rad)	1.76375702	=RADIANS(C23)					
25	9	dec/lat 2 (deg)	-16.68638889	=DMSDD(C12,C13,C14)					
26	10	dec/lat 2 (rad)	-0.291232426	=RADIANS(C25)					
27	11	cos(*d*)	0.915845952	=SIN(C21)*SIN(C26)+COS(C21)*COS(C26)*COS(C19-C24)					
28	12	*d* (rad)	0.413186619	=ACOS(C27)					
29	13	*d* (deg)	23.67384942	=DEGREES(C28)					

Figure 34. Finding the angle between two celestial objects.

Thus you could delete rows 17 to 29 of the spreadsheet shown in Figure 34 (having saved a copy), and insert into cells H3, H4 and H5 the following formulas:

```
=DDDeg(Angle(C3,C4,C5,C6,C7,C8,C9,C10,C11,C12,C13,C14,C15))
=DDMin(Angle(C3,C4,C5,C6,C7,C8,C9,C10,C11,C12,C13,C14,C15))
=DDSec(Angle(C3,C4,C5,C6,C7,C8,C9,C10,C11,C12,C13,C14,C15)).
```

33 Rising and setting

During the course of a sidereal day, the stars and other ‘fixed’ celestial objects appear to move in circles about the rotation axis of the Earth, making one complete revolution in 24 hours. At the moment, there is a star called Polaris very close to the north pole of the Earth’s axis so that stars in the northern sky appear to revolve about Polaris. There is nothing special about this star, however, and no corresponding object exists for the south pole. In any case, the poles are gradually changing their positions in the sky because of precession (see the next section) so that Polaris will no longer be the pole star in a few thousand years.

The apparent radius of a star's rotation depends, of course, on the angular separation, or **polar distance**, between it and the pole; those stars with a small enough polar distance never dip below the horizon during the course of their rotation. Such stars are called **circumpolar**. As the polar distance increases, however, a point comes when the star just touches the horizon at some time during the day. Stars with polar distances greater than this spend part of their time below the horizon, out of sight to the observer. When the star crosses the horizon on the way down it is said to **set** and as it reappears it is said to **rise**.

There are several effects, including atmospheric refraction (Section 37) and parallax (for bodies relatively close to the Earth: Sections 38 and 39), that shift an object's apparent position and this may alter the apparent times of rising or setting by several minutes. The situation at rising or setting is shown in Figure 35. The celestial body appears to cross the horizon at B, although its 'true' position, as calculated from its uncorrected coordinates, is at A. Provided we know the vertical shift[†], v, we can include its effects on the circumstances of rising and setting.

The local sidereal times of rising and setting, and the azimuths at which they occur, can be calculated using the formulas

$$\cos H = -\frac{(\sin v + \sin\phi \sin\delta)}{\cos\phi\cos\delta},$$

$$\mathrm{LST_r} = \alpha - H,$$

$$\mathrm{LST_s} = \alpha + H,$$

$$\cos A_r = \frac{\sin\delta + \sin v \sin\phi}{\cos v \cos\phi},$$

$$A_s = 360° - A_r,$$



Figure 35. The true and apparent positions of a celestial object at rising or setting.

[†] v is positive if the star stays longer above the horizon.

where the subscripts r and s correspond to rising and setting respectively, A is the azimuth, LST is the local sidereal time in hours, α is the right ascension, δ is the declination, ϕ is the observer's geographical latitude and H is the hour angle. The value of $\cos H$ can be used as an indicator of whether the star never rises, or is circumpolar. If $\cos H$ is greater than 1, the star is permanently below the horizon and never rises. If $\cos H$ is more negative than -1, the star is permanently above the horizon and never sets (i.e. is circumpolar).

The LST can be converted to UT and hence to the local civil time by the methods given in Sections 15, 13 and 10. Hence, all the circumstances of a star's rising and setting can be calculated. However, there is a difficulty that you may need to overcome if you live far away from the Greenwich meridian. In order to convert the Greenwich sidereal time into the UT, you need to know the calendar date at Greenwich on which the rising or setting occurs. But in order to find this from your local calendar date you need to know the UT at which the rising or setting occurs. This difficulty is easily overcome however. If you take the date at Greenwich to be the same as your local calendar date, the times of rising and setting will usually not be more than a few minutes out. You can then use those times to recalculate the calendar date(s) at Greenwich and iterate until there are no further changes.

As an example, let us calculate the UTs of rising and setting over a sea horizon of a star whose equatorial coordinates are α = 23h 39m 20s and $\delta = 21° \, 42' \, 00''$ on 24 August 2010, and find the corresponding azimuths. The geographical latitude is 30° N, and the longitude is 64° E, and the value of ν due to atmospheric refraction is 34 arcmin.

Method	*Example*		
1. Convert α and δ into decimal form (§§7 and 21).	α δ	= =	23.655 558 hours 21.700 000 degrees
2. Find $\cos H = -\frac{(\sin\nu + \sin\phi\sin\delta)}{\cos\phi\cos\delta}$.	$\cos H$	=	−0.242 047
3. If $\cos H$ is between -1 and $+1$, take the inverse cos to find[a] H.	H	=	6.933 827 hours
4. Find $\mathrm{LST_r} = \alpha - H$. Restore to the range 0 to 24 by adding or subtracting 24.	$\mathrm{LST_r}$	=	16.721 728 hours
5. Find $\mathrm{LST_s} = \alpha + H$. Restore to the range 0 to 24 by adding or subtracting 24.	$\mathrm{LST_s}$	=	6.589 383 hours
6. Find $A_r = \cos^{-1}\left\{\frac{\sin\delta + \sin\nu\sin\phi}{\cos\nu\cos\phi}\right\}$. Restore to the range 0 to 360 by adding or subtracting 360.	A_r	=	**64.362 348 degrees**
7. Find $A_s = 360 - A_r$.	A_s	=	**296.637 652 degrees**
8. Convert the LST values to GST values, then to universal times (§§15 and 13).	$\mathrm{UT_r}$ $\mathrm{UT_s}$	= =	14.271 670 hours 4.166 990 hours
9. Finally, express the times as hours, minutes and seconds (§8).	$\mathrm{UT_r}$ $\mathrm{UT_s}$	= =	**14h 16m** **4h 10m**

[a]If the star's declination is such that it never rises above the horizon, or if it is circumpolar, then you will find that you will be trying to take inverse cos of a number greater than 1 or less than -1. This is impossible and your calculator should respond with 'error'.

Note that the UTs you calculate are appropriate for the date you have applied. As here, the setting time on a given date may be earlier than the rising time.

Figure 36 shows the spreadsheet for making this lengthy calculation. We have used several techniques that are worthy of note. First, in rows 29 and 30, we have used the trick of adding 30 s (= 0.008 333 hours) to the UTs so that, when displayed as hours and minutes, the time will be rounded correctly to the nearest



	A	B	C	D–F	G	H	I
1	**Correcting for Aberration**						
2							
3	*Input*	**UT (hour)**	**0**	*Output*	**apparent ecl long (deg)**	**352**	=DDDeg(C21)
4		**UT (min)**	**0**		**apparent ecl long (min)**	**37**	=DDMin(C21)
5		**UT (sec)**	**0**		**apparent ecl long (sec)**	**30.45**	=DDSec(C21)
6		**G date (day)**	**8**		**apparent ecl lat (deg)**	**-1**	=DDDeg(C22)
7		**G date (month)**	**9**		**apparent ecl lat (min)**	**32**	=DDMin(C22)
8		**G date (year)**	**1988**		**apparent ecl lat (sec)**	**56.33**	=DDSec(C22)
9		**true ecl long (deg)**	**352**				
10		**true ecl long (min)**	**37**				
11		**true ecl long (sec)**	**10.1**				
12		**true ecl lat (deg)**	**-1**				
13		**true ecl lat (min)**	**32**				
14		**true ecl lat (sec)**	**56.4**				
15							
16	1	true long (deg)	352.6194722	=DMSDD(C9,C10,C11)			
17	2	true lat (deg)	-1.549	=DMSDD(C12,C13,C14)			
18	3	Sun true long (deg)	165.5633044	=Sunlong(C3,C4,C5,0,0,C6,C7,C8)			
19	4	dlong (arcsec)	20.35217443	=-20.5*COS(RADIANS(C18-C16))/COS(RADIANS(C17))			
20	5	dlat (arcsec)	0.068073433	=-20.5*SIN(RADIANS(C18-C16))*SIN(RADIANS(C17))			
21	6	apparent long (deg)	352.6251256	=C16+(C19/3600)			
22	7	apparent lat (deg)	-1.548981091	=C17+(C20/3600)			

Figure 39. Correcting ecliptic coordinates for the effects of aberration.

37 Refraction

In all our calculations so far, we have assumed that the light from distant objects reaches us by the most direct route, a straight line. This is not actually the case (except for observations made at the zenith) as the Earth's atmosphere bends the light a little, making the rays reach the ground at a slightly different angle from that which they would have had if the atmosphere had not been there (see Figure 40). This is called **atmospheric refraction** and its effect is to make the star appear to be closer to the zenith than it really is. The amount of refraction depends on the **zenith angle** or **zenith distance** (90° − altitude) and on the atmospheric conditions, particularly the temperature and pressure. If we observe a star with zenith angle ζ from the surface of the Earth, its true zenith angle, z, is given by $z = \zeta + R$, where R is the refraction angle. An approximate expression for R that is suitable for altitudes above 15° is

$$R = 0.00452 P \tan z / (273 + T) \text{ degrees,}$$

where T is the temperature in degrees centigrade and P is the barometric pressure in millibars, both measured at the observation point. This formula is usually accurate to about 6 arcsec for altitudes greater than 15°. At lower altitudes, better results can be obtained using the approximate formula

$$R = \frac{P\left(0.1594 + 0.0196a + 0.00002a^2\right)}{(273+T)\left(1 + 0.505a + 0.0845a^2\right)} \text{ degrees,}$$

where a is the altitude in degrees.†

†Strictly, a is the apparent altitude as measured through the atmosphere, rather than the true altitude as measured with no atmosphere.



Figure 40. Atmospheric refraction.

The effect of refraction on true equatorial, ecliptic and galactic coordinates is best computed by first converting to horizon coordinates, increasing the altitude by adding R, and then converting back to the original coordinate system to find the apparent position. We will now illustrate this by calculating the refraction for a star whose true hour angle is 5h 51m 44s and true declination +23° 13′ 10″ as observed at a geographical latitude of 52° N. The temperature is 13° C and the pressure is 1008 mbar.

Method	*Example*		
1. Calculate the true altitude and azimuth of the star (§25).	a	=	19.334 345 degrees
	A	=	283.271 027 degrees
2. Find the refraction angle R from the formula appropriate to the altitude: $a > 15°$, $R = 0.00452P \tan z/(273+T)$.	z	=	70.665 655 degrees
	R	=	0.045 403 degrees
3. Add R to the altitude to find the apparent altitude a'.	a'	=	19.379 748 degrees
4. Convert A and a' back into equatorial coordinates (§26).	H'	=	**5h 51m 36s**
	δ'	=	**23° 15′ 14″**

The magnitude of R right at the horizon is usually assumed to be 34 arcmin. (Its actual value may be different depending on atmospheric conditions.) Since its effect is to increase the apparent altitude, the times of rising and setting will be earlier and later, respectively, than they would have been without the atmosphere. The effective length of the day, therefore, is increased by atmospheric refraction. We can calculate its effects on the azimuths and times of rising and setting by the method given in Section 33. Alternatively, we can calculate the effect on the hour angle, H, at rising or setting by

$$\Delta H = \frac{34}{15 \cos\phi \cos\delta \sin H} \text{ minutes of time,}$$

where ΔH is the amount by which the true hour angle is reduced.

	A	B	C	D	G	H	I
1	**Selenographic coordinates 2**						
2							
3	*Input*	Greenwich date (day)	1		*Output* sub-solar longitude	6.81	=ROUND(C27,2)
4		Greenwich date (month)	5		sub-solar colongitude	83.19	=ROUND(C28,2)
5		Greenwich date (year)	1988		sub-solar latitude	1.19	=ROUND(C22,2)
6							
7	1	Julian date (days)	2447282.5	=CDJD(C3,C4,C5)			
8	2	*T* (centuries)	-0.11670089	=(C7-2451545)/36525			
9	3	long asc node (deg)	350.7599447	=125.044522-1934.136261*C8			
10	4	*F*	-56296.83349	=93.27191+483202.0175*C8			
11	5	*F*	223.1665139	=C10-360*INT(C10/360)			
12	6	Sun geocentric long (deg)	40.84263343	=SunLong(0,0,0,0,0,C3,C4,C5)			
13	7	Moon equ hor parallax (arc min)	55.95238522	=MoonHP(0,0,0,0,0,C3,C4,C5)*60			
14	8	Sun-Earth dist (AU)	1.00760326	=SunDist(0,0,0,0,0,C3,C4,C5)			
15	9	geocentric Moon lat (rad)	-0.053838535	=RADIANS(MoonLat(0,0,0,0,0,C3,C4,C5))			
16	10	geocentric Moon long (deg)	209.1175282	=MoonLong(0,0,0,0,0,C3,C4,C5)			
17	11	adjusted Moon long (deg)	220.7476113	=C12+180+(26.4*COS(C15)*SIN(RADIANS(C12-C16))/(C13*C14))			
18	12	adjusted Moon lat (rad)	-0.000140054	=0.14666*C15/(C13*C14)			
19	13	inclination (rad)	0.026920249	=RADIANS(DMSdd(1,32,32.7))			
20	14	node-long (rad)	2.269143285	=RADIANS(C9-C17)			
21	15	sin(bs)	0.020755895	=-COS(C19)*SIN(C18)+SIN(C19)*COS(C18)*SIN(C20)			
22	16	sub-solar lat (deg)	1.189310602	=DEGREES(ASIN(C21))			
23	17	*A* (rad)	-2.269324173	=ATAN2(COS(C18)*COS(C20),-SIN(C18)*SIN(C19)-COS(C18)*COS(C19)*SIN(C20))			
24	18	*A* (deg)	-130.0226975	=DEGREES(C23)			
25	19	sub-solar long (deg)	-353.1892113	=C24-C11			
26	20	sub-solar long (deg)	6.810788676	=C25-360*INT(C25/360)			
27	21	sub-solar long (deg)	6.810788676	=IF(C26>180,C26-360,C26)			
28	22	sub-solar colong (deg)	83.18921132	=90-C27			

Figure 51. Calculating the selenographic coordinates of the Sun.

43 Atmospheric extinction

The light that reaches us on the surface of the Earth from heavenly bodies first has to pass through the atmosphere where some of it is scattered by dust, electrons, oxygen and nitrogen molecules, and other sundry particles. The amount of this **Rayleigh scattering** depends on the physical conditions in the atmosphere (it will be enhanced, for example, by extra dust from a volcanic eruption) and on the wavelength of the light. In general, the shorter wavelengths (blue) are scattered much more than the longer wavelengths (red); for this reason, the sky looks blue (we see the scattered light) and the apparent colour of a star observed from the Earth's surface is reddened. If we take the visual wavelengths as a whole, we can make a rough estimate of the amount of absorption to expect when the atmosphere is clear, from

$$\Delta m = \frac{0.2}{\cos z} \text{ magnitudes,}$$

where Δm is the quantity to be added to the **magnitude**, and z is the zenith angle ($z = 90^\circ -$ altitude). For example, a planet whose altitude is 15° may appear dimmer by about 0.8 magnitudes in good conditions when the atmosphere is clear; in general this will be an underestimate since there are additional causes of absorption. The formula breaks down for zenith angles greater than about 85°.



Figure 56. Finding the position of the Sun by a more precise method.

48 Calculating the Sun's distance and angular size

Having found the true anomaly, ν, by the method of Sections 46 or 47, we can easily calculate the Sun–Earth distance, r, and the Sun's angular size (i.e. its angular diameter), θ. The formulas are:

$$r = r_0 \left(\frac{1 - e^2}{1 + e\cos\nu} \right),$$

$$\theta = \theta_0 \left(\frac{1 + e\cos\nu}{1 - e^2} \right),$$

where r_0 is the semi-major axis, θ_0 is the angular diameter when $r = r_0$, and e is the eccentricity of the orbit. These constants are given in Table 7. Continuing the example of Section 47 we can find r and θ for the Sun on Greenwich date 27 July 1988 at 0 h UT.

Method	*Example*		
1. Find the true anomaly, ν (§§46 or 47).	ν	=	201.443 110 degrees
2. Find $f = \frac{1+e\cos\nu}{(1-e^2)}$. (See Table 7 for e.)	f	=	0.984 726
3. Then $r = \frac{r_0}{f}$. (See Table 7 for r_0.)	r	=	**$1.519\,189 \times 10^8$ km**
4. And $\theta = f\theta_0$. (See Table 7 for θ_0.)	θ	=	**$0^\circ\ 31'\ 30''$**

The *Astronomical Almanac* gives $\theta = 0^\circ\ 31'\ 30''$ and, in general, we should be within a few arcseconds of the correct value. It is interesting to note that the Sun's light took r/c seconds to reach us, where $c = 3 \times 10^5\ \mathrm{km\ s^{-1}}$. In this case the light travel time was 506 seconds, during which interval the Sun moved

	T_p (tropical years)	ε (degrees)	ϖ (degrees)	e	a (AU)	i (degrees)	Ω (degrees)	θ_0 (arcsec)	V_0
Mercury	0.240 85	75.567 1	77.612	0.205 627	0.387 098	7.005 1	48.449	6.74	−0.42
Venus	0.615 207	272.300 44	131.54	0.006 812	0.723 329	3.394 7	76.769	16.92	−4.40
Earth	0.999 996	99.556 772	103.205 5	0.016 671	0.999 985				
Mars	1.880 765	109.096 46	336.217	0.093 348	1.523 689	1.849 7	49.632	9.36	−1.52
Jupiter	11.857 911	337.917 132	14.663 3	0.048 907	5.202 78	1.303 5	100.595	196.74	−9.40
Saturn	29.310 579	172.398 316	89.567	0.053 853	9.511 34	2.487 3	113.752	165.60	−8.88
Uranus	84.039 492	271.063 148	172.884 833	0.046 321	19.218 14	0.773 059	73.926 961	65.80	−7.19
Neptune	165.845 39	326.895 127	23.07	0.010 483	30.198 5	1.767 3	131.879	62.20	−6.87

1 AU = 149.6×10^6 km.
T_p: period of orbit; ε: longitude at the epoch; ϖ: longitude of the perihelion; e: eccentricity of the orbit; a: semi-major axis of the orbit; i: orbital inclination; Ω: longitude of the ascending node; θ_0: angular diameter at 1 AU; V_0: visual magnitude at 1 AU.

Table 8. Elements of the planetary orbits at epoch 2010.0.

Comet name	P	ϖ (degrees)	Ω (degrees)	T_p (years)	a (AU)	e	i (degrees)
Encke	1974.32	160.1	334.2	3.30	2.209	0.847	12.0
Temple 2	1972.87	310.2	119.3	5.26	3.024	0.549	12.5
Haneda–Campos	1978.77	12.016	131.700	5.37	3.066	0.641 52	5.805
Schwassmann–Wachmann 2	1974.70	123.3	126.0	6.51	3.489	0.386	3.7
Borrelly	1974.36	67.8	75.1	6.76	3.576	0.632	30.2
Whipple	1970.77	18.2	188.4	7.47	3.821	0.351	10.2
Oterma	1958.44	150.0	155.1	7.88	3.958	0.144	4.0
Schaumasse	1960.29	138.1	86.2	8.18	4.054	0.705	12.0
Comas Sola	1969.83	102.9	62.8	8.55	4.182	0.577	13.4
Schwassmann–Wachmann 1	1974.12	334.1	319.6	15.03	6.087	0.105	9.7
Neujmin 1	1966.94	334.0	347.2	17.93	6.858	0.775	15.0
Crommelin	1956.82	86.4	250.4	27.89	9.173	0.919	28.9
Olbers	1956.46	150.0	85.4	69.47	16.843	0.930	44.6
Pons–Brooks	1954.39	94.2	255.2	70.98	17.200	0.955	74.2
Halley	1986.112	170.011 0	58.154 0	76.008 1	17.943 5	0.967 3	162.238 4

P: epoch of the perihelion; ϖ: longitude of the perihelion; Ω: longitude of the ascending node; T_p: period of the orbit; a: semi-major axis of the orbit; e: eccentricity of the orbit; i: inclination of the orbit.

Table 9. The orbital elements of some periodic comets.